
In this chapter, we discuss at length how to write programs that work as Web clients. The World Wide Web
is a ubiquitous application that sits atop the Internet which is a diffused network of networks encompassing
the globe. It came into existence in 1990 with the first Web server and Web client developed by Berners-Lee.
The first client was textual whereas the dominant browsers or clients at this time, viz., Netscape Navigator
and Microsoft Internet Explorer are GUI-based.
As we are well-aware, the World Wide Web employs the client-server model of computing. We discuss
client-server computing at length in Chapter 7. The dominant servers at this time are the Apache Server,
and Microsoft’s IIS Server. Other severs include AppleShare IP that runs on Macintosh computers. Web
servers are extremely complex programs that serve Web pages to Web clients. Commercial Web clients are
extremely complex programs as well. As a result, we do not venture into writing of complex clients, but
simple ones that can fetch Web pages and perform useful computation on these Web pages. By definition,
a Web client communicates with Web servers. A Web server usually listens on port 80 on the machine that
hosts it. Web clients and servers communicate using the HTTP protocol. A protocol defines a language’s
syntax as well as imposes constraints on what can be said when. Web pages are written mostly using the
HTML language although increasingly the language called XML is finding wider acceptance. In addition,
when dynamic Web pages are served by a Web server, the Web server does not respond directly to a request
from a client, but obtains responses by talking to an intermediary. This intermediary can take various forms
such as a CGI program, an ASP program, a PHP program, or a JSP program.
In this section, we discuss how to write programs that can send a request to a Web server, get a response
back, and perform useful computation with the page that is returned. Unlike a commercial Web browser, we

do not format the page returned to make it pretty, or deal with graphics, audio or video. However, we write
programs that can be immediately useful. For example, a comparison-shopping program that presents a
table of prices for a specific product from competing on-line stores performs Web client programming. A
program that monitors various on-line auction sites performs Web client programming. A program that
obtains news stories from several Web sites and presents the stories best suited to an individual’s tastes
performs Web client programming.
The communication takes place between a Web server and a Web client uses the HTTP protocol that sits
atop the TCP protocol we discuss in Chapter 7. The TCP protocol sits on top of the IP protocol which is
discussed at length in any book on computer networking. Thus, Web clients and servers must understand
TCP and IP. A Web client such as a program that fetches a single Web page, or crawls the Web fetching many
relevant pages, can be written employing sockets that use the TCP protocol. We wrote such programs in
Chapter 7. However, Perl makes writing Web clients much easier than starting from scratch using TCP-
based sockets. It has a set of related modules for developing fairly sophisticated Web clients. We discuss
the most important such modules in this section.

9.1 Fetching a Web Page: Module LWP::Simple

There is a large set of Perl modules that are bundled together and are called the Library for (World Wide)
Web Programming or LIBWWW or LWP. It is also sometimes called libnet.
One of the simplest modules is LWP::Simple. As the name suggests, it is a subclass of the module LWP.
If the goal at hand is to fetch a Web page and do something with it, this package is all that we need. The
following program fetches a Web page, and saves it to a local file. It also prints all the URLs that are in the
page and were used as a hyperlink using the <A href=...> construct in HTML.

Program 9.1

#!/usr/bin/perl
#file fetchHeadURL.pl
use LWP::Simple;
use strict;

#URL to fetch
my $url = "http://www.assam.org";
my $localFile = "assamorg.html";

my @head = LWP::Simple::head ($url);
exit 0 unless (@head);
my ($contentType, $documentLength,

$modificationTime, $expires, $server) = @head;
print "$url is a file of type $contentType\n";
print "$url is $documentLength bytes long\n";

print "$url was last modified on ", scalar (localtime($modificationTime)), "\n";
print "$url expires on $expires\n" if ($expires);
print "$url is served by $server\n";

#get gets the URL as a string
my $content= LWP::Simple::get ($url);
my @hrefs = ($content =˜ /<A\s+href\s*=\s*"([ˆ"]+)"/ig);
print "All hrefs in the page = \n\t@hrefs\n";

open FILE, ">$localFile";
print FILE $content;
close FILE;

The program is given a URL to fetch and a local file where to save the fetched URL. First, it fetches the
head of the page. The LWP::Simple::head function returns a list with five elements. The elements are
extracted out of the list and printed. The first element gives the MIME type for the content of the Web
page. The content type is sent by the Web server to a client or a browser in the header so that the client
can decide how to display it. In this example, the content type is text/html signifying that it is a text file
written using the syntax of the HTML language. The second element gives the length of the page in bytes.
The third element gives the time when the page was last modified in terms of number of seconds after the
so-called epoch. The epoch is system-specific. For Unix, it is January 1, 1970. In other words, it is the age
of the page. The expiry date can be empty for a Web page. The server gives details of the server software
and hardware. Next, the program gets the actual content of the URL by using the LWP::Simple::get
function. While head obtains the header from the Web server, get obtains the actual Web page. Each
takes a URL as an argument. These two functions automatically set up TCP-based socket connections to
the designated Web servers, send the HTTP-based commands as needed, capture the data that come back
to the client, and then save it in appropriate data structures. Because connections need to be set up and
communications need to take place, these functions usually take some time in coming back with responses.
The program culls out all hrefed URLs from the page. This part of the program is based on the discussion
in Section 4.8.3.2.3. It also prints out the contents of the URL to a file. Such a program can be the basis for a
sophisticated crawler or a search engine program.
A partial output of one run of the program is given below.

http://www.assam.org is a file of type text/html
http://www.assam.org is 25141 bytes long
http://www.assam.org was last modified on Thu Jan 25 02:07:54 2001
http://www.assam.org is served by Apache/1.3.14 (Unix)

(Red-Hat/Linux) ApacheJServ/1.1.2 mod_ssl/2.7.1
OpenSSL/0.9.6 PHP/4.0.2 mod_perl/1.21

All hrefs in the page =
http://www.assamcompany.com
mailto:kalita@pikespeak.uccs.edu,webmaster@assam.org
http://assamcompany.com/netourism/start.html
http://mail.bigmailbox.com/users/assamorg/signup.cgi
http://mail.bigmailbox.com/users/assamorg/forgotpassword.cgi
chat/livechat.htm
http://assam.org/assam/AssamBulletinBoard/
http://www.assam.org/assam/individuals/
http://www.wunderground.com/global/stations/42314.html
http://www.wunderground.com/global/stations/42410.html

The output continues with more such URLs.

9.1.1 Fetching Documents from the Web: Simple “Web Crawling”

There are many sites on the Web where useful documents are stored so that interested individuals can
download them. For example, the site
http://www.cis.upenn.edu/˜techreports/abstracts01.html at the University of Pennsylva-
nia contains abstracts of all technical reports published by the Computer Science Department in the year

2001. Most of these abstracts have links to PostScript or PDF files that are downloadable from the Web. The
following program fetches the HTML page corresponding to the URL given above, obtains the links to the
technical reports, and then downloads the technical reports to local files.

Program 9.2

#!/usr/bin/perl
#file getFiles.pl
use LWP::Simple;

my ($ftpURL, $ftpContents, $ftpFile);
my ($htmlURL, $ftpHrefs, $includedExts);
$includedExts = "ps[.]gz|pdf|ps";
$htmlURL = "http://www.cis.upenn.edu/˜techreports/abstracts01.html";

$htmlContents = LWP::Simple::get ($htmlURL) or
die "Couldn’t fetch $htmlURL using LWP::Simple::get: $!";

@ftpHrefs = ($htmlContents =˜ /<A\s+href\s*=\s*"([ˆ"]+)"/ig);
@ftpHrefs = grep /[.]$includedExts$/, @ftpHrefs;
#print "ftp hrefs = @ftpHrefs\n";

foreach $ftpURL (@ftpHrefs){
print "Fetching URL: $ftpURL\n";
($ftpFile) = ($ftpURL =˜ m@.+/(.+)$@);
print "Saving to local file: $ftpFile\n";
$ftpContents = LWP::Simple::get ($ftpURL);
warn "Couldn’t fetch $ftpURL: $!" unless $ftpContents;
open FTPFILE, ">$ftpFile";
print FTPFILE $ftpContents;
close FTPFILE;

}

The program does not really use the FTP protocol although the names of identifiers has the term FTP in
them. The HTML file corresponding to abstracts is first fetched using LWP::Simple::get. The hrefed
URLs are next culled from the HTML file. From the URLs, only those that end with the required extension
are kept. The required extensions in this program are ps, pdf, or ps.gz. In other words, these are files
in the Adobe PostScript format or PDF format. The PostScript files may have been compressed using the
Gnu zip command (i.e., gzip). Such compressed files have the gz extension. We know that these are the
relevant extensions by examining the HTML page. Our goal is to download all technical reports to our local
machine so that we can peruse them either on the computer screen or after printing them.
The program goes into a foreach loop where it obtains the name of the actual file from the URL. The name
of the file is the component in the URL after the last /. Then, it obtains the contents of the file by making
the following call.

$ftpContents = LWP::Simple::get ($ftpURL);

The contents of the downloaded file are written into the local file. The downloading is performed by issuing
the GET command to the HTTP server. That is what the get function does. A part of what is printed by
this program on the screen for a sample run is given below.

Fetching URL: http://www.cis.upenn.edu/˜rtg/papers/MS-CIS-01-01.ps.gz

Saving to local file: MS-CIS-01-01.ps.gz
Fetching URL: http://www.cis.upenn.edu/˜lwzhao/papers/cognitive.pdf
Saving to local file: cognitive.pdf
Fetching URL: http://www.cis.upenn.edu/˜sotiris/papers/strongman.ps
Saving to local file: strongman.ps
Fetching URL: http://www.cis.upenn.edu/˜sotiris/papers/subos.ps
Saving to local file: subos.ps
Fetching URL: http://www.cis.upenn.edu/˜rtg/papers/MS-CIS-01-07.ps.gz
Saving to local file: MS-CIS-01-07.ps.gz
Fetching URL: http://www.seas.upenn.edu/˜sachinc/report.ps
Saving to local file: report.ps
Fetching URL: http://www.cis.upenn.edu/˜pengs/publications/fc_tech_report.ps.gz
Saving to local file: fc_tech_report.ps.gz
Fetching URL: http://db.cis.upenn.edu/DL/ubql.pdf
Saving to local file: ubql.pdf

In this sample run, the program downloads eight technical reports to the local machine. The downloading
is done not using the FTP protocol, but by using the HTTP protocol. This could have been done using the
FTP protocol also, but we do not discuss that here. In this specific case, it so happens that all the URLs
referring to downloaded files are absolute URLs. Therefore, we do not have to massage the URL to make it
absolute before we start the downloading process. If the URLs were relative, it would be necessary to make
them absolute first before starting to download.
A variation of the program given above can be used to download Perl modules from the
http://www.cpan.org site. The following program looks at the Web page that lists the recently con-
tributed Perl modules and downloads those that satisfy a certain regular expression.

Program 9.3

#!/usr/bin/perl
#file PerlModules.pl
use LWP::Simple;

my ($ftpURL, $ftpContents, $ftpFile);
my ($htmlURL, $ftpHrefs, $includedExts, $baseURL);
my ($searchPattern) = @ARGV;

$includedExts = "tar[.]gz";
$htmlURL = "http://www.cpan.org/RECENT.html";
$baseURL = "http://www.cpan.org";

$htmlContents = LWP::Simple::get ($htmlURL) or
die "Couldn’t fetch $htmlURL using LWP::Simple::get: $!";

@ftpHrefs = ($htmlContents =˜ /<A\s+href\s*=\s*"([ˆ"]+)"/ig);
@ftpHrefs = grep /[.]$includedExts$/, @ftpHrefs;
@ftpHrefs = grep /$searchPattern/, @ftpHrefs;
@ftpHrefs = map {"$baseURL/$_"} @ftpHrefs;

foreach $ftpURL (@ftpHrefs){
print "Fetching URL: $ftpURL\n";
($ftpFile) = ($ftpURL =˜ m@.+/(.+)$@);
print "Saving to local file: $ftpFile\n";

$ftpContents = LWP::Simple::get ($ftpURL);
warn "Couldn’t fetch $ftpURL: $!" unless $ftpContents;
open FTPFILE, ">$ftpFile";
print FTPFILE $ftpContents;
close FTPFILE;

}

The program is called with a command-line argument that acts as a regular expression. This program
obtains the Web page at http:://www.cpan.org/RECENT.html, parses the page to obtain all URLs
that end with the tar.gz extension. Then, it obtains those URLs that satisfy the regular expression given
as command line argument. Finally, it downloads all the tarred sources of recent Perl modules that satisfy
the search criterion. An interaction with this program is given below.

%getPerlModules.pl "XML"
Fetching URL: http://www.cpan.org/authors/id/A/AN/ANDREIN/DBIx-XMLMessage-0.04.t
ar.gz
Saving to local file: DBIx-XMLMessage-0.04.tar.gz
Fetching URL: http://www.cpan.org/authors/id/K/KR/KRAEHE/XML-Handler-YAWriter-0.
18.tar.gz
Saving to local file: XML-Handler-YAWriter-0.18.tar.gz
Fetching URL: http://www.cpan.org/authors/id/K/KR/KRAEHE/XML-Handler-YAWriter-0.
19.tar.gz
Saving to local file: XML-Handler-YAWriter-0.19.tar.gz
Fetching URL: http://www.cpan.org/modules/by-category/07_Database_Interfaces/DBI
x/DBIx-XMLMessage-0.04.tar.gz
Saving to local file: DBIx-XMLMessage-0.04.tar.gz
Fetching URL: http://www.cpan.org/modules/by-category/11_String_Lang_Text_Proc/X
ML/XML-Handler-YAWriter-0.19.tar.gz
Saving to local file: XML-Handler-YAWriter-0.19.tar.gz
Fetching URL: http://www.cpan.org/modules/by-module/DBIx/DBIx-XMLMessage-0.04.ta
r.gz
Saving to local file: DBIx-XMLMessage-0.04.tar.gz
Fetching URL: http://www.cpan.org/modules/by-module/XML/XML-Handler-YAWriter-0.1
9.tar.gz
Saving to local file: XML-Handler-YAWriter-0.19.tar.gz
pikespeak[98]: ls

The program is called with "XML" as the command line argument. Hence, the program finds those Perl
modules that have been recently contributed to the http://www.cpan.org site and have the character
string XML in them, and then downloads them to the local machine. These are compressed tar files corre-
sponding to the source programs for these modules. One can then go about installing these modules on the
local computer if so desired.

9.1.2 Filling a GET form on the Web: Automatically Finding Book Prices

Web pages are full of forms. A form is a way to interact with a Web server and get a response back. For
example, we fill in a form to become a member of a mailing list, or to search for items using a search engine
or an electronic store. A simple form can have only one or two elements in them whereas a complex form
can have many elements to fill.
There are two ways that a filled form can be submitted to a Web server requesting an action on its part and
a subsequent response. The two methods are called GET and PUT. The GET method sends all the form’s
filled contents to the server in one network transmission whereas the PUT method needs an additional
transmission, one to send the form’s URL and headers, and another to send the form’s data. The form’s

Figure 9.1: A Search Form at www.borders.com

data are accepted by the server and then processed by a program such as a CGI program and the response
is sent back to the user either via the Web server or directly.
The designer of a form decides what method between the two to use. For simple forms, GET is sufficient.
The differences between the two methods can be found in any book that discusses HTML and the HTTP
protocol. Mimicking a form submitted with PUT is a little difficult to handle. However, GET-submitted
forms can be treated as a regular URL as we see below.
The following program fills in a form at the Web site for the electronic book and music store called
http:://www.borders.com. The site has a simple form that allows a visitor to search for books using
an ISBN number. This form is at the location
http://search.borders.com/fcgi-bin/db2www/search/search.d2w/BookISBN.
Before discussing the program, it must be noted that Web sites change frequently, and therefore, the form
being discussed in the book may not exist at a future date. The form, as it looks on the Web site at the time
of writing, is shown in Figure 9.1.
On the top left, we see a form that allows a visitor to search for book, music or video/DVD titles. There
is a second form that allows search using a ten-digit ISBN number. If we look at the HTML for the second
form, we see a form that looks like the following.

<TD BGCOLOR="#ffffcc"><FORM NAME="ISBNUPC_form" METHOD="get" action="Details">
<B>Enter ISBN to look for:</B><SMALL><BR>&nbsp;<BR></SMALL>
<INPUT TYPE="text" name="code" SIZE="30" MAXLENGTH="100">
<INPUT TYPE="submit" VALUE="Find ISBN">
<INPUT TYPE="hidden" NAME="mediaType" VALUE="Book">
<INPUT TYPE="hidden" NAME="searchType" VALUE="ISBNUPC">
<INPUT TYPE="hidden" NAME="prodID" VALUE="">
<P><B>Instructions:</B><BR>
Enter a book’s 10-digit ISBN number. Do not use spaces.</P>

</TD>
</FORM>

Finding the correct form in the source of the Web page may take a little time. The form is inside an HTML

table’s cell represented by the tag TD. This form’s ACTION attribute refers to the URL Details. The form’s
METHOD attribute has the value get. Note that HTML is case-insensitive in the way we specify tags such as
FORM, and their attributes such as ACTION. Thus, the form submission takes place using the GET method
of the HTTP protocol. The form has several hidden arguments: code, mediaType, searchType and
prodID. They all have values specified in HTML. These values are sent to the server when the form is
submitted by clicking on the button labeled Find ISBN. The code text field is the one we see in the form
in the Web browser. In the program that follows, we automatically fill in this form and submit it to the
Web server. The Web server performs the search and returns an HTML-formatted page with the result of
the search. We examine this page carefully and see that the price of the book is always preceded by certain
keywords. We key in on this repeated occurrence and parse the returned page to obtain the price of the
book, and print it on the screen.

Program 9.4

#!/usr/bin/perl
#file bordersISBN1.pl

use strict;
use LWP::Simple;

my ($url, $content);
my $ISBN = "1565922433";
print "ISBN = $ISBN\n";

#Make up the URL to search for the book’s ISBN
$url = "http://search.borders.com/fcgi-bin/db2www/search/search.d2w/Details?";
$url .= "code=$ISBN&mediaType=Book&searchType=ISBNUPC";

$content = LWP::Simple::get ($url);
my ($price) = ($content =˜ m#Our Price:.+?\$(.+?)<#si);
print "price = $price\n";

The program uses the module LWP::Simple. It is given an ISBN number that happens to be 1565922433.
Of course, this ISBN number could have been given as a command-line argument or read from the terminal
after an appropriate prompt. The program creates a URL to request the server to initiate a search. To
perform a search using a GET-submitted form, the server is sent the URL followed by a question mark, and
then by one or more form field name, form field value pairs. The name and the value are separated by
an equal sign. Different pairs are separated by the ampersand (&). Note that values must be supplied to
hidden form elements with the correct values as found in the Web page for the form. This is the manner in
which GET submitted forms are sent by a browser to a server. In this case, the URL sent to the server is the
following. We have broken it into two lines whereas it is sent as a single line to the server.

http://search.borders.com/fcgi-bin/db2www/search/search.d2w/Details?
code=$ISBN&mediaType=Book&searchType=ISBNUPC

There are no intervening spaces. The search URL seems to work without a prodID value.
This program simply mimics what a Web browser does. When the form is received by the Web server, it
does the needful to get the search performed, and then returns the results of the search. If we were actually
doing this search using a Web browser, we see the result page that looks like the one given in Figure 9.2. We
carefully look at the HTML of this result page and result pages of several additional searches, and find that

Figure 9.2: A Search Form at www.borders.com

every time we perform an ISBN search, the price of the book charged by the store follows the keywords
Our Price. We obtain the price by simply parsing the Web page’s HTML. Note that if we wanted to find
additional information such as the title of the book, author, shipping information, etc., we will have to do
additional search of the Web page’s HTML. The output of running this program is given below.

%bordersISBN1.pl
ISBN = 1565922433
price = 35.96

This printout says that the books with ISBN number 1565922433 is sold by www.borders.com at a price of
$35.96. This program can be easily extended to read a sequence of ISBN numbers from a file or a database,
and obtain prices for each one of the ISBN numbers. If prices can be fetched from several bookstores, we
can very easily compare prices of books.

9.2 The LWP Bundle:: Writing Sophisticated Web Clients

The LWP::Simple module discussed in the previous section provides a simple functional interface for
simple Web-client related tasks. However, to write sophisticated Web clients, we need more muscle. The
set of modules that constitute the bundle called LWP provides additional facilities.
To write powerful Web clients, we need to be familiar with the HTTP protocol or the Hypertext Transfer
Protocol. It is the language that Web clients and servers use to communicate with each other. A request from
a client to a server and a subsequent response from the server to the client constitute an HTTP transaction.
A client request and a server response follow the same syntax. There is a request or response line followed
by a header section and then the entire body.
A client always initiates a transaction by using the following steps.

1. The client contacts the server at the port number used by the Web server. The default port number is
80.
Once the connection has been established, the client sends a request to the server. The request is
framed in terms of an HTTP command, usually called a method. The method is given arguments such
as the address of the document to be fetched, and an HTTP version number.
The number of HTTP requests or methods is limited and they are discussed in books on networking,
or in a site such as www.w3.org, the World Wide Web Consortium. The commonly used HTTP
methods are GET, HEAD, POST and PUT. The GET method is used by a client to request a server to
send a document found at a specific location. It can be used to fill forms that use the GET action
attribute. The HEAD method is similar to GET, but it requests only some information on a file or a
resource, and not the actual document. The POST method allows data to be sent to the server in a
client request. For example, it can be used to provide data to a server for a newsgroup posting, or for
a database operation. The POST method is frequently used by forms instead of the GET method. The
PUT method is normally used to publish a document on a Web site.

2. Next, the client may send some header information to the server. The header information tells the
server of the client’s configuration and document types that the client can accept. The header infor-
mation is a sequence of lines, each line containing a header name followed by a header value. The
header information is optional.

3. Finally, the client may optionally send additional data. This additional data is usually meant for POST
forms. It may also contain the content of a file to be published by the server.

The server gets the client’s request at the specified port and performs the needed action and then responds
in the following manner.

1. The first line is the status line containing the HTTP version, the status code, and a description of the
status such as the word OK.

2. The next several lines are header information sent by the server to the client. The header information
usually contains information about the server itself and information about the requested document.
The header lines are terminated by a blank line.

3. If the server is successful in fulfilling the client’s request, the requested data is sent next.

When a client sends an HTTP request to a server, if the server is alive and can respond to the request, it
does so. As far as client programming goes, we do not have to worry about what the server does and how.
The client however needs to capture the response that comes back from the server and deal with it. If the
client requests a header, a header comes back and Perl captures it. If a file is requested, Perl gets a header
back as well as the contents of the file.
Perl provides two object-oriented modules: HTTP::Request and HTTP::Response to model an HTTP
interaction at the client’s end. These are used to create a request to send and capture the response that comes
back. A program creates an HTTP::Request object, and sends it. The response is captured automatically
as an HTTP::Response object. To send out an HTTP::Request and to receive the response, Perl provides
a class called LWP::UserAgent. The user agent is a software entity that actively pursues Web client related
activities on behalf of a user or an application. It is a conduit between an application and a Web server. In
simple and informal terms, it is an incarnation of the Web client. It takes an HTTP request, say to fetch a
Web page, and then waits for the response to arrive from the server. Once the response is received, the user
agent makes it available to the rest of the program.
In the simple use situation, an application creates a LWP::UserAgent object, and creates an
HTTP::Request object for the request that needs to be performed. The request is then given to the
request method of the UserAgent object. The request method opens up a communication channel
with the server and sends the request out. When the response comes back, the response is captured by Perl
in the form of a HTTP::Response object. Figure 9.3 describes the process diagrammatically.

Figure 9.3: The LWP::UserAgent object on the client sends an HTTP::Request object to the server. The
UserAgent object receives the reply as an HTTP::Response object.

The request() method of the UserAgent object can make arrangements to get the response in one of
three ways: as a scalar available for direct manipulation, to be saved in a file, or handed over to a callback
routine for processing. The first alternative is useful if the returned content is small and needs additional
processing such as parsing to cull out relevant data. The second alternative is suitable for large objects such
as a graphic or audio file. The last alternative can be used to process data in chunks as it comes in.

9.2.1 Requesting URL Header From a Web Server

We start with a program that communicates with a Web server requesting the header information for a file.

Program 9.5

#!/usr/bin/perl
#file fetchHead1.pl

use LWP::UserAgent;
use HTTP::Request;
use URI;
use strict;
$" = "\n\t";

my ($url, $uri, $ua);
my ($headerRequest, $headerResponse, $headers);

#URL to fetch
$url = "http://www.cs.uccs.edu/˜kalita";
$uri = URI->new($url);

#Creating a user agent and sending a request, and getting response
$ua = LWP::UserAgent->new();
$headerRequest = HTTP::Request->new(HEAD=>$uri);

$headerResponse = $ua->request ($headerRequest);
if ($headerResponse->is_success){

$headers = $headerResponse->headers;
}
else{

print $headerResponse->error_as_HTML;
exit 0;

}

print "-" x 60, "\n";
print "THE HEADER RESPONSE IS...\n",

$headerResponse->as_string, "\n";
print "-" x 60, "\n";
print "status line: ", $headerResponse->status_line, " status\n";
print "-" x 60, "\n";

print "HEADERS AS STRING IS...\n", $headers->as_string, "\n";
print "-" x 60, "\n";
print "content type: ", $headers->content_type, "\n";
print "date: ", $headers->date, "\n";
print "server: ", $headers->server, "\n";

printf "%-15s %-30s\n", "Header Name", "Header Value";
print "-" x 60, "\n";
$headers->scan (\&headerScanner);
print "-" x 60, "\n";

#callback subroutine to process header entries
sub headerScanner{

my ($headerName, $headerValue) = @_;
printf "%-15s %-30s\n", $headerName, $headerValue;

}

The program uses the LWP::UserAgent and HTTP::Request modules. It also uses a module called URI
to create a Uniform Resource Identifier for the URL to fetch. It is not really necessary to create a URI out of a
URL to use it. We do so for illustration purposes only.
The program is given a value for a URL it fetches. A URL is given in the form of a string. The program
converts the URL to an URI object. A URI is a more generalized abstraction of an address than a URL.
However, the differences are not important to us at this time.

$url = "http://www.cs.uccs.edu/˜kalita";
$uri = URI->new($url);

As mentioned earlier, it is not necessary to convert the URL to a URI, a more generalized address, for the
purpose of this program. Next, the program creates a user agent object using the following line.

$ua = LWP::UserAgent->new();

The program also creates an HTTP request and passes this request to the user agent.

$headerRequest = HTTP::Request->new(HEAD=>$uri);

The request is for header information, and takes the URI created earlier as an argument. The argument
could have been a URL in string form as well. Next, the request is given to the user agent’s request
method.

$headerResponse = $ua->request ($headerRequest);

The user agent’s request method opens a TCP-based communication socket with the server. If successful,
the user agent converts the request to the proper HTTP format and sends it to the server over the socket.
We discussed sockets in detail in Chapter 7. The LWP bundle of modules makes it unnecessary to perform
low-level tasks such as creating sockets for Web client programming, making things simple for the pro-
grammer. The user agent waits for the response to come back from the server. When the response comes,
the user agent composes an HTTP::Response object out of it automatically. Thus, $headerResponse
is an instance of the HTTP::Response class, whether the request was successful in fetching the header
or not. The HTTP::Response class has two methods is_success and headers. If the HTTP request
is successful in obtaining the HEAD information for the URI, the is_success returns true. If the request
fails, the program prints the error sent to it by the server in HTML format and exits. The error_as_HTML
method of the HTTP::Response class does so.
The HTTP::Response object has an internal format for storage in the program. To print the content of the
object in string form, we can use the as_string method provided by the HTTP::Response object. If we
look at the output printed below, we see that the statement below

print "THE HEADER RESPONSE IS...\n",
$headerResponse->as_string, "\n";

prints the complete response that comes back from the server. The information printed by the statement
is given later in this discussion. The first line is the HTTP response line we discussed earlier. It says the
protocol used is HTTP/1.1. The request was successful and came back with a 200 status code. The status
is given as OK in text form. Following this, there is a list of HTTP header and value pairs. The headers
listed here are Connection, Date, Server, Content-Length, etc. The value of $headerResponse
when printed as a string contains one or more blank lines at the end although the HTTP specification
says there should be only one blank line. Thus, the HTTP::Response class has several useful methods
such as is_success, is_error, as_string, error_as_HTML, etc. Another method that we use in this
program is status_line that prints just the status line or the first line of the response. Two other useful
methods that the HTTP::Response object has are called headers and content. Every HTTP request
comes back with a response that contains headers. However, for the HEAD HTTP request, no content comes
back. Thus, we capture only the headers in the response if the request is successful. The capturing is done
in the program inside the if-else statement.

$headers = $headerResponse->headers;

The headers method of the HTTP::Response class returns an object of yet another class called the
HTTP::Headers class that is also available as a part of the LWP bundle of packages. When sending out an
HTTP Request, one can optionally create headers to be sent out with the request. On the flip side, when a
response has come back, the headers that have come back can be captured for examination.
The HTTP::Headers class also has several methods that are quite useful in examining the contents of the
headers. Examples of these methods are content_type, date, and server used in this program. One
interesting method that HTTP::Headers class has is called scan. A call to scan is given below.

$headers->scan (\&headerScanner);

The argument taken by scan is a reference to a function. The function must take two arguments, the name
of a header and the value of a header. The definition of such a function used in the program is given below.

sub headerScanner{
my ($headerName, $headerValue) = @_;
printf "%-15s %-30s\n", $headerName, $headerValue;

}

A function such as this is called a callback function. A callback function is applied to every header in turn.
It is called with a header name and a single value. It can do whatever it pleases with the arguments. In this
case, it simply prints them in the form of a formatted string. Note that if a header has several values, the
function is called once for each value. The output of this program is given below.

------------------------------------------------------------
THE HEADER RESPONSE IS...
HTTP/1.1 200 OK
Connection: close
Date: Wed, 04 Apr 2001 20:01:12 GMT
Accept-Ranges: bytes
Server: Apache/1.3.14 (Unix) (Red-Hat/Linux) PHP/3.0.18 mod_perl/1.23
Content-Length: 4225
Content-Type: text/html
ETag: "650328-1081-3ab64734"
Last-Modified: Mon, 19 Mar 2001 17:51:48 GMT
Client-Date: Wed, 04 Apr 2001 13:04:48 GMT
Client-Peer: 128.198.162.68:80

------------------------------------------------------------
status line: 200 OK status
------------------------------------------------------------
HEADERS AS STRING IS...
Connection: close
Date: Wed, 04 Apr 2001 20:01:12 GMT
Accept-Ranges: bytes
Server: Apache/1.3.14 (Unix) (Red-Hat/Linux) PHP/3.0.18 mod_perl/1.23
Content-Length: 4225
Content-Type: text/html
ETag: "650328-1081-3ab64734"
Last-Modified: Mon, 19 Mar 2001 17:51:48 GMT
Client-Date: Wed, 04 Apr 2001 13:04:48 GMT
Client-Peer: 128.198.162.68:80

------------------------------------------------------------
content type: text/html
date: 986414472
server: Apache/1.3.14 (Unix) (Red-Hat/Linux) PHP/3.0.18 mod_perl/1.23
Header Name Header Value
------------------------------------------------------------
Connection close
Date Wed, 04 Apr 2001 20:01:12 GMT
Accept-Ranges bytes
Server Apache/1.3.14 (Unix) (Red-Hat/Linux) PHP/3.0.18 mod_perl/1.23

Content-Length 4225
Content-Type text/html
ETag "650328-1081-3ab64734"
Last-Modified Mon, 19 Mar 2001 17:51:48 GMT
Client-Date Wed, 04 Apr 2001 13:04:48 GMT
Client-Peer 128.198.162.68:80
------------------------------------------------------------

9.2.2 Obtaining a URL From a Web Server

The next program makes a call to the HTTP method HEAD to obtain information on a URL. If the call to
HEAD returns successfully, only then the program obtains the actual file. The program follows.

Program 9.6

#!/usr/bin/perl
#file fetchHeadURL1.pl

use LWP::UserAgent;
use URI;
use strict;
$" = "\n\t";

my ($url, $uri, $ua);
my ($localFile);
my ($headerRequest, $headerResponse);
my ($contentRequest, $contentResponse, $headers, $content, @hrefs);

$url = "http://www.cs.uccs.edu/˜kalita";
$uri = URI->new($url);
$ua = LWP::UserAgent->new();

$localFile = "assamorg.html";

$headerRequest = HTTP::Request->new(HEAD=>$uri);
$headerResponse = $ua->request ($headerRequest);
unless ($headerResponse->is_success){

print $headerResponse->error_as_HTML;
exit 0;

}

$contentRequest = HTTP::Request->new(GET=>$uri);
$contentResponse = $ua->request ($contentRequest);
$headers = $contentResponse->headers;
$content = $contentResponse->content;

printf "%-15s %-30s\n", "Header Name", "Header Value";
print "-" x 60, "\n";
$headers->scan (\&headerScanner);
print "-" x 60, "\n";

#get gets the URL as a string
@hrefs = ($content =˜ /<A\s+href\s*=\s*"([ˆ"]+)"/ig);
print "All hrefs in the page = \n\t@hrefs\n";

open FILE, ">$localFile";
print FILE $content;
close FILE;

#callback subroutine to process header entries
sub headerScanner{

my ($headerName, $headerValue) = @_;
printf "%-15s %-30s\n", $headerName, $headerValue;

}

Like the previous program, this program also is given a URL which it converts to a URI object. It creates a
LWP::UserAgent object and sends out an HTTP request for the HEAD of a URI using the requestmethod
of the LWP::UserAgent object. If the HEAD request is unsuccessful, the program prints the error and exits.
If the HEAD request is successful, it makes up a GET request, sends it over and then captures the response.
The following lines of code do that.

$contentRequest = HTTP::Request->new(GET=>$uri);
$contentResponse = $ua->request ($contentRequest);
$headers = $contentResponse->headers;
$content = $contentResponse->content;

The GET method requests a URI on the server. As usual, the request is sent to the server by using the
request method of the LWP::UserAgent object. The response that comes back from the server is called
is
$contentResponse. The response captured by the user agent is automatically an HTTP::Response
object. We call the headers method to capture the header part of the response. Unlike the previous
program, this time we also call the contentmethod to capture the content in the variable $content. In the
previous program, we fetched only the HEAD from a Web server and hence, the content part of the response
was empty. But, this time, the content part contains the actual text of the file requested. The headers
method returns an object of type HTTP::Headers whereas the content returns the actual content of the
requested URL. The content can be textual or binary.
Before we look at the requested URI’s contents, we look at the HTTP headers that came back with the file.
HTTP headers always accompany a response from a server. We make a call to the scan method of the
HTTP::Headers class to print the contents of the header lines one by one. Next, we look at the actual
content of the file. Since the file requested is small in size, we handle it directly. This program uses a regular
expression to cull all URLs used in the file fetched and print them on the screen. The program also prints
the contents of the file fetched into a local file, thus mirroring the Web page locally.
The output of the program is given below.

Header Name Header Value
------------------------------------------------------------
Connection close
Date Wed, 04 Apr 2001 20:03:52 GMT
Accept-Ranges bytes
Server Apache/1.3.14 (Unix) (Red-Hat/Linux) PHP/3.0.18 mod_perl/1.23
Content-Length 4225
Content-Type text/html

ETag "650328-1081-3ab64734"
Last-Modified Mon, 19 Mar 2001 17:51:48 GMT
Client-Date Wed, 04 Apr 2001 13:07:28 GMT
Client-Peer 128.198.162.68:80
Link <default.css>; rel="stylesheet"; type="text/css"
Title Jugal Kalita
------------------------------------------------------------
All hrefs in the page =
xml/index.xml
teaching-philosophy.pdf
schedule.html
research.html
http://www.shillong.com
http://www.autoindia.com
http://www.indiashipping.com
http://www.assam.org
http://www.assamcompany.com
cultural.html
http://www.upenn.edu/index.html
http://www.usask.ca
http://www.rahul.net/kgpnet/iit/iit.html
http://www.uccs.edu
http://www.cs.uccs.edu/˜kalita/accesswatch/accesswatch-1.32/index.html

It is not really necessary to issue the HEAD request to gather information about a page before issuing the
GET command to fetch the page. One can simply use the GET command bypassing the HEAD command.
The HEAD command is used only if our goal is to find out certain information about the file such as its last
modification date before deciding whether to fetch the file or not. This could be useful if one is trying to
create a local mirror of a large Web site. Only files modified after the last time mirroring was done need
to be fetched, possibly reducing the amount of fetching to be done to a great extent. Fetching the head or
the whole file both need one transmission over the network. But, the head is a lot less information than the
whole file, and thus, the amount of transmission as well computation, both at the server end and the client
end is reduced quite a bit.
The following program fetches a URL without fetching the head information first. The program also culls
out all the hrefed URLs specified in the file, just like the previous program.

Program 9.7

#!/usr/bin/perl
#file fetchURL3.pl

use LWP::UserAgent;
use URI;
use strict;
$" = "\n\t";

my ($url, $uri, $ua);
my ($localFile);
my ($contentRequest, $contentResponse, $headers, $content, @hrefs);

$localFile = "assamorg.html";

$url = "http://www.cs.uccs.edu/˜kalita";
$uri = URI->new($url);
$ua = LWP::UserAgent->new();

$contentRequest = HTTP::Request->new(GET=>$uri);
$contentResponse = $ua->request ($contentRequest);
if ($contentResponse->is_success){

$content = $contentResponse->content;
}
else{

print $contentResponse->error_as_HTML;
exit 0;

}

#get gets the URL as a string
@hrefs = ($content =˜ /<A\s+href\s*=\s*"([ˆ"]+)"/ig);
print "All hrefs in the page = \n\t@hrefs\n";

open FILE, ">$localFile";
print FILE $content;
close FILE;

The program sends out at GET method using the request method of the user agent. When the response
comes back, it uses the is_success method of the HTTP::Response object to see if the request was
successfully processed by the server. If the response contained an error, the program exits. Otherwise, it
fetches the content from the response, culls out all the hrefed URLs, like before. It also prints the fetched
file to a local mirror file.

9.2.3 Automatically Filling a GET Form Using HTTP::Request

In section 9.1.2, we see how a form that uses the GET method can be automatically filled using the
LWP::Simple module. We see in that section that filling a GET form and obtaining response is very sim-
ple. Of course, the same can be done with the more complex LWP modules such as LWP::UserAgent,
HTTP::Request, HTTP::Response and others.
A GET form request is sent to the Web server in a single transmission as a URL. That is why processing
GET forms is easy. The following program is similar in objective and functionality to the program described
in 9.1.2.

Program 9.8

#!/usr/bin/perl
#file bordersISBN2.pl

use strict;
use HTTP::Request;
use LWP::UserAgent;

my ($url, $content);
my ($ua, $contentRequest, $contentResponse, $price);
my $ISBN = "1565922433";

print "ISBN = $ISBN\n";

#Make up the URL to search for the book’s ISBN
$url = "http://search.borders.com/fcgi-bin/db2www/search/search.d2w/Details?";
$url .= "code=$ISBN&mediaType=Book&searchType=ISBNUPC";

$ua = LWP::UserAgent->new();
$contentRequest = HTTP::Request->new(GET=>$url);
$contentResponse = $ua->request ($contentRequest);
if ($contentResponse->is_success){

$content = $contentResponse->content;
}
else{

print $contentResponse->error_as_HTML;
exit 0;

}
($price) = ($content =˜ m#Our Price:.+?\$(.+?)<#si);
print "price = $price\n";

The program is almost exactly like the programs we see in Section 9.2. The program uses the
LWP::UserAgent module to create a new user agent called $ua. The request is formed in terms of a URL
where the form parameters follow the ? mark following the URL. Although we have a statement

use HTTP::Request;

on the top of the program, it is not necessary to use the HTTP::Requestmodule. The form is automatically
submitted by creating an HTTP::Request object with the URL as the argument. Note that, in general, it is
not necessary to convert the URL to a URI before sending it to a Web server, as we have done in examples
in previous sections.
The response that comes back from the server is captured by the user agent $ua. The response is automat-
ically an HTTP::Response object. If the response has a successful status code, the HTML content of the
content is saved in the variable $content. This variable is later parsed to obtain the price of the book with
the ISBN number given in the variable $ISBN.
There is another module called HTTP::Request::Common that makes it slightly more convenient to write
common HTTP requests. For GET forms, we need to simply set up a user agent and then issue a GET request
as given below.

$contentRequest = GET $url;

Thus, it is as simple as the LWP::Simple module to use. The GET $url call automatically sets up the
appropriate HTTP::Request object. It also captures the response coming back from the Web server auto-
matically. A program given earlier in this section can be rewritten as shown below.

Program 9.9

#!/usr/bin/perl
#file bordersISBN4.pl

use strict;
use HTTP::Request::Common;
use LWP::UserAgent;

my ($url, $content);
my ($ua, $contentRequest, $contentResponse, $price);
my $ISBN = "1565922433";
print "ISBN = $ISBN\n";

#Make up the URL to search for the book’s ISBN
$url = "http://search.borders.com/fcgi-bin/db2www/search/search.d2w/Details?";
$url .= "code=$ISBN&mediaType=Book&searchType=ISBNUPC";

$ua = LWP::UserAgent->new();
$contentRequest = GET $url;
$contentResponse = $ua->request ($contentRequest);
if ($contentResponse->is_success){

$content = $contentResponse->content;
}
else{

print $contentResponse->error_as_HTML;
exit 0;

}
($price) = ($content =˜ m#Our Price:.+?\$(.+?)<#si);
print "price = $price\n";

9.2.4 Automatically Filling a POST Form

We have seen how forms that use the GET method can be submitted automatically. We can do so with forms
submitted using the POST method as well, although it takes a little more work to do so. The following
program finds the price of a book given the ISBN number at an Internet bookstore called powells.com.

Program 9.10

#!/usr/bin/perl
#file powellISBN1.pl

use strict;
#use LWP::Debug qw(+);
use HTTP::Response;
use HTTP::Request;
use LWP::UserAgent;

my ($ua, $response, $url);
my ($content, $req);
my $searchFor = "1565922433";
my $searchType = "ISBN";
print "ISBN = $searchFor\n";

$ua = LWP::UserAgent->new();
$ua->agent ("Mozilla/6.0; compatible");
$ua->timeout(200);

442



On Perl 9.2. THE LWP BUNDLE:: WRITING SOPHISTICATED WEB CLIENTS

$url = "http://www.powells.com/search/DTSearch/search";
$req = HTTP::Request->new(POST=>$url);
$req->content(qq{isbn=$searchFor});

$response = $ua->request($req);
$content = $response->content();

########################************************************
###NEEDS MORE WORK. IT’S PICKING UP THE FIRST PRICE
#ONLY, BUT POWELLS.COM MAY RETURN SEVERAL ENTRIES FOR THE SAME BOOK.
#########################*******************
my ($price) = ($content =˜ m#RESULT ITEM START.+?\$(.+?)<#si);
print "price = $price\n";

The program creates a user agent $ua and gives the user agent a couple of attributes. Using the agent
method, it identifies itself as compatible to Mozilla/6.0 type browsers. It also sets a timeout of 200
seconds to get the result back. It then specifies the URL of the form’s action attribute in terms of the
variable $url. This URL has to be found by examining the source of the appropriate Web page where
the form for searching book prices at powells.com is situated. We determine that the HTML source of the
form specifies that it is using the POST method. The GET method is default if nothing is specified. An
HTTP::Request object is created using the URL as the argument for a POST form. The HTTP::Request
object’s new method takes one or more arguments in the form of hash fields. One such hash field’s name
is POST and the corresponding value is the URL. The HTTP::Request object $req uses the content
method to specify the form element names and their values. In this case, there is only one relevant form
attribute and it is called isbn. Its value is given as the value of the variable $searchFor. qq is the quoting
function that allows interpolation of variable values. It is like using double quotes to delimit a string. If
there were several fields, each field’s name and value are written separated from the next pair by a comma.
The user agent uses the request method of the LWP::UserAgent object to send the HTTP request to the
Web server. The Web server responds and the response is the result of the request method. The content
method of the HTTP::Response object gives the content of the response, as opposed to the header. The
response is very simply parsed to obtain the price of the book with the given ISBN number. Once again, to
find out where the price occurs in the returned Web page, we need to fill in the form manually several times
and see if we can determine a simple way to figure out where the price occurs in the returned result. In this
specific case, the price always occurs after the phrase RESULT ITEM START. However, in this particular
store powells.com, the search for a single book can return several prices. Here, only the first one is captured.
We have seen in section 9.2.3 how the HTTP::Request::Common module can be used to process GET
forms. The HTTP::Request::Common module can be used to submit POST forms as well. The following
program is a rewrite of the program given earlier in this section.

Program 9.11

#!/usr/bin/perl
#file powellISBN2.pl

use strict;
#use LWP::Debug qw(+);
use HTTP::Response;
use LWP::UserAgent;
use HTTP::Request::Common;

my ($ua, $response, $url);
my ($content, $req);
my $searchFor = "1565922433";
my $searchType = "ISBN";
print "ISBN = $searchFor\n";

$ua = LWP::UserAgent->new();
$ua->agent ("Mozilla/5.0");
$ua->timeout(600);

$url = "http://www.powells.com/search/DTSearch/search";
$req = POST "$url", ["isbn" => $searchFor];
$response = $ua->request($req);
$content = $response->content();

my ($price) = ($content =˜ m#RESULT ITEM START.+?\$(.+?)<#si);
print "price = $price\n";

In this program, the statement that sends out the POST form is given below.

$req = POST "$url", ["isbn" => $searchFor];

There is only one relevant field in the form. If there were several fields, the field name, value pairs are
separated from each other by commas. Of course, instead of having the URL defined earlier, we could have
specified the URL with the POST call directly, as given below.

$req = POST "http://www.powells.com/search/DTSearch/search",
["isbn" => $searchFor];

9.3 Using Cookies

Cookies are mechanism by which a Web server can store information on a Web client. This information can
be later retrieved by the Web server. Cookies are used frequently by many Web servers to keep track of who
comes and visits the sites. The first time someone comes to a site, the server can send one or more cookies
to the client. The client stores the cookies in a local file, if the client is enabled to do so. Many Web servers
do not allow a client to visit the site unless cookies are enabled. Some sites allow only partial access if the
client is not enabled to accept cookies.
The following program is somewhat complex. It is a small part of a much bigger program, written by
the author and others, that fetches news items from various on-line sources, puts them together, removes
duplicates, classifies them automatically, and then presents the news items to every individual that visits
the Web site as a personalized newspaper. The details are fairly involved using sophisticated techniques
based on artificial intelligence based learning.
The program presented here logs onto the site apnewstracker.com using a login name and a password.
The top page of the site at a certain instant of time is shown in Figure 9.4. If the login is unsuccessful, it
attempts to login several times, waiting a specified number of seconds after each attempt. The number of
times is obtained from a separate file that we call a configuration file. The site sends back cookies to the
browser and the browser needs to save them locally to be able to proceed further with retrieving the news
items posted on the site. If the login is successful, the site automatically returns a page containing an index

Figure 9.4: The Top Page at www.apnewstracker.com

of news items with a headline for each item and a link to a URL containing the text of the item. The index
page for a particular instant of time is shown in Figure 9.5. The program parses the index page sent to it,
and culls the URLs of the individual news item. There are several parts to the index page and the program
picks out the news items in the first part only. The first part is indicated by the presence of certain specific
syntactical items in the HTML of the index Web page. Next, the program fetches each one of the news
items. A partial Web page for a specific news item is shown in Figure 9.6.
To fetch each news item URL, it has to present cookies to the Web server along with the request to fetch
the page. In the more sophisticated program not discussed here, the program parses each news item’s
text, saves it in some form in a database, classifies it into pre-determined categories, and performs much
additional computation to be useful. We do not present these other processing steps in this simplified
version of the program since these are irrelevant to our current discussion.
The program’s text is given below, followed by more explanation.

Program 9.12

#!/usr/bin/perl
#file apget1.pl

use strict;
#use LWP::Debug qw (+);
use HTTP::Response;
use LWP::UserAgent;
use HTTP::Cookies;
use HTTP::Request::Common;
use Time::Local;
use AP_config;

###Main Program ##################################
##########Global declarations. These are declared in a configuration file####

my ($cookie_jar, $ua, $loginResponse, $newsItemIndex);
my (%newsIndexHash);
my ($itemURL, $itemText);

#Create cookie_jar to be stored in file $AP_config::cookieFile
$cookie_jar = HTTP::Cookies -> new (file => $AP_config::cookieFile,

ignore_discard => 1);

#define the user agent

Figure 9.5: The News Index Page at www.apnewstracker.com

Figure 9.6: A News Item at www.apnewstracker.com

$ua = LWP::UserAgent -> new ();
$ua -> agent ($AP_config::userAgentDescription);
$ua -> timeout ($AP_config::userAgentTimeout); #default is 180 seconds

#Do for ever. Login+fetch news items every
#$$AP_config::waitTimeBetweenFetches seconds
while (1){

my $noLoginTriesThisTime = $AP_config::noLoginTries;

LOGIN:{
print "\nTrying to login to $AP_config::loginURL...\n\n";
#log into the AP site and Fetch the news item index
$loginResponse = &login($ua, $cookie_jar);
print "After login\n";

if (!$loginResponse){
print "Couldn’t login\n";
$noLoginTriesThisTime--;

#Reduce the number of times login will be tried sequentially
if ($noLoginTriesThisTime){

print "SLEEPING $AP_config::waitTimeBetweenFailedLogins seconds...\n";
sleep ($AP_config::waitTimeBetweenFailedLogins);
goto LOGIN; #Try to login again

}else{
print "Tried logging in $AP_config::noLoginTries, didn’t succeed, exiting...\n";
exit 1;

}
}

} #LOGIN block ends

print "Just logged onto the http://www.apnewstracker.com site...\n";

for (my $i = 1; $i <= $AP_config::repeatFetchTimesBetweenLogins; $i++){
#logged in at this point
$newsItemIndex = &fetchIndexPage ($ua, $cookie_jar);

#Parse the news index page that comes back; if there is an HTTP error,
#$newsItemIndex is the number 0
%newsIndexHash = ();
if (!$newsItemIndex){

print "Couldn’t fetch news items index page...\n";
print "SLEEPING $AP_config::waitTimeBetweenFailedLogins";
print "seconds before retrying login...\n";
sleep ($AP_config::waitTimeBetweenFailedLogins);
goto LOGIN; #Try to login again

}

&parseIndexPage ($newsItemIndex); #This puts values in %newsIndexHash

#Next, need to fetch the URLs specified in the index page
#But, need to fetch only those URLs that are not already there in cache
foreach $itemURL (reverse (sort (keys %newsIndexHash))){

$itemText = &fetchNewsItemURL ($ua, $cookie_jar, $itemURL);
if (!$itemText){

print "Could not fetch url: $itemURL\n";
}

} #foreach ends

#sleep for a while
print "\nSLEEPING FOR $AP_config::waitTimeBetweenFetches SECONDS...\n\n";
sleep ($AP_config::waitTimeBetweenFetches);

} #for my $i ends, fetched news indices
#$AP_config::repeatFetchTimesBetweenLogin times

} #while (1) ends

#########SUBROUTINES####################################
################sub login################################
#login to the AP newstracker page. This page is at
#$AP_config::baseURL = http://www.newstracker.com.
#The form to fill in to login is at $AP_config::loginURL
sub login{

my ($ua, $cookie_jar) = @_;

my ($request, $response);
my ($indexContents);

#add cookie information to the request
#the request logs one the apnewstracker.com site
$request = POST $AP_config::loginURL,

[username => $AP_config::loginName, password => $AP_config::loginPassword];
print $request->as_string . "\n";
$cookie_jar -> add_cookie_header ($request);

#Get the response to the request and extract cookies from what comes back
$response = $ua -> request ($request);
$cookie_jar -> extract_cookies ($response);

#print HTTP error message if the user couldn’t login or there was an HTTP error
if ($response -> is_success){

return $response;
} else {

print "Could not login to the http://www.apnewstracker.com site\n";
print $response -> error_as_HTML;
return 0; #If it couldn’t login or there was an HTTP error, it returns the number 0

}
} #sub login ends

################sub fetchIndexPage################################
#Fetch the index page of news items
sub fetchIndexPage{

my ($ua, $cookie_jar) = @_;
#print "Inside fetchIndexPage subroutine\n";

my ($request, $response);
my ($indexContents);

#Now, get the actual page where the news index is.
#This page http://www.apnewstracker.com/is3/runprofile.hts is updated
#every minute or two.
$request = GET $AP_config::newsIndexURL;
$cookie_jar -> add_cookie_header ($request);
$response = $ua -> request ($request);
$cookie_jar -> extract_cookies ($response);

#Parse the news index page if it comes back. If there
#is HTTP error, print an error page
if ($response -> is_success){

print "Obtained $AP_config::newsIndexURL\n";
$indexContents = $response -> content;
return $indexContents;

} else {
print $response -> error_as_HTML;
return 0; #return 0 if there is a HTTP error

}
} #sub fetchIndexPage ends

################sub parseIndexPage####################################
#A subroutine that parses the AP News index page:
# http://www.apnewstracker.com/is3/runprofile.hts
#The page is passed to the subroutine as a string

sub parseIndexPage{
my ($fileContents) = @_;
my (@tables, @items);
my ($item, $month, $day, $year, $url, $title, $hour, $minute, $headline);

my ($relevantPart) =
($fileContents =˜ m#DataStream A</FONT>(.+?)technology<#s);

@items = ($relevantPart =˜ m#(<TABLE.+?</TABLE>)#gsi);

foreach $item (@items){
#The sequencing of the substitutions and pattern matches matter
($month, $day, $year) = ($item =˜ m#(\d{2})/(\d{2})/(\d{2})#);

($url, $title) = ($item =˜ m#<a href="([ˆ"]+).+?>(.+?)</a>#);
$item =˜ s#<a.+?</a>##gsi; #obtained the URL, now remove <a...>...</a>

#Removing all the things below to clean things up.

#Otherwise, difficult to parse
$item =˜ s#<!--.+?-->##gs; #remove comments
$item =˜ s#[\s]+# #gs; #remove multiple spaces
$item =˜ s#</?table.+?>##gsi;
$item =˜ s#</?tr.+?>##gsi;
$item =˜ s#</?td.+?>##gsi;
$item =˜ s/&nbsp;/ /gs; #substitute an &nbsp by white space
$item =˜ s#<font.+?>##gsi;
$item =˜ s#</font>##gi;

($hour, $minute) = ($item =˜ /(\d{2}):(\d{2})/);
($headline) = ($item =˜ /$hour:$minute(.+)/sg);
($headline) = ($headline =˜ /ˆ\s*(.+)\s*/sg); #remove spaces from front and end
#Sometimes there are HTML tags in headline, remove them
$headline =˜ s#<.+?>##gs;

print "Parsing index item...\n";
$year = "21$year";
print "time = $month/$day/$year $hour:$minute\n";
print "title = $title\n";
print "headline = $headline\n\n";

my $itemData = join ":::",
($year, $month, $day, $hour, $minute, $title, $headline);

$newsIndexHash{$AP_config::baseURL.$url} = $itemData;
} #foreach ends

} #sub parseIndexPage ends

###############sub fetchNewsItemURL #########################
###A subroutine that fetches a news item URL ############
sub fetchNewsItemURL{

my ($ua, $cookie_jar, $url) = @_;
my $request = GET $url;
$cookie_jar -> add_cookie_header ($request);
my $response = $ua -> request ($request);

$cookie_jar -> extract_cookies ($response);

if ($response -> is_success){
my $newsItemText = $response -> content;
return $newsItemText;

}else{
print $response -> error_as_HTML;
print "Couldn’t fetch news item URL: $url\n";
return 0;

}
}

The program uses several modules we have seen earlier: HTTP::Response, LWP::UserAgent,
HTTP::Cookies and HTTP::Request::Common. These are pre-defined well-regarded modules that can
be downloaded from the Internet. We also have a module called AP_config that declares and gives values
to certain variables used in the program. The module is shown below. The login name and the password
have been replaced by a sequence of Xs.

Program 9.13

package AP_config;
#Define all site-specific global variables
use strict;
use vars qw($baseURL $loginURL $newsIndexURL);
use vars qw($loginName $loginPassword); #To login to apnewstracker.com
use vars qw($waitTimeBetweenFetches $waitTimeBetweenFailedLogins);
use vars qw($repeatFetchTimesBetweenLogins $noLoginTries);
use vars qw($userAgent $userAgentTimeout);

$baseURL = "http://www.apnewstracker.com";

$loginURL = "$baseURL" . "/is3/wsh-kickoff.hts";
$newsIndexURL = "$baseURL" . "/is3/runprofile.hts";
$loginName = "XXXXXXX";
$loginPassword = "XXXXXX";

#Global variables for database access
$waitTimeBetweenFetches = 180;
$waitTimeBetweenFailedLogins = 180;
$noLoginTries = 10;
$repeatFetchTimesBetweenLogins = 20;
$userAgent = "Mozilla/4.7 (Compatible; MSIE 5.0; Windows2000)";
$userAgentTimeout =400;

1;

One of the first things that the main program does is to specify a file to store cookies sent by a Web server.

#Create cookie_jar to be stored in file $AP_config::cookieFile
$cookie_jar = HTTP::Cookies -> new (file => $AP_config::cookieFile,

ignore_discard => 1);

This is done by creating a new HTTP::Cookies object. The call to the new object constructor method takes
one or more parameters, given in the form of a hash. The file argument gives the name of the file where
cookies are stored when they are sent to the browser by the Web server. The second argument is boolean. It
is not necessary that we use it like we do here. This parameter instructs that the program save even cookies
that are requested to be destroyed by the server. Initially, the cookie file is empty.
Next, the program creates a user agent $ua, gives a description to the agent using the agent method and
specifies a time after which the user agent should give up by using the timeout method. Once the setup is
over, the program goes into a while loop where it attempts to log in the the www.apnewstracker.com
site. The logging is done using the login subroutine that takes the user agent and the cookie file, called
$cookie_jar, as arguments. If the login is not successful, the program sleeps for a pre-determined
amount of time and retries. It attempts to login only a pre-specified number of times, say 10. If even after
all attempts, the program does not succeed, it dies. This can happen because of reasons such as that the
server is down or is overloaded, or the server is unreachable.
Once the program has logged in to the site, it fetches the index page a specified number of times after which
it logs in again. This makes the program robust because it was found that the site logged the program out
after a certain amount of time. The way the program is written now, it can go for many days fetching news
items without any problems. The fetching of the news index page is done using the following statement.

#logged in at this point
$newsItemIndex = &fetchIndexPage ($ua, $cookie_jar);

If the news item index page cannot be fetched, it generally means that the program has been logged out.
Therefore, in such a case, the program tries to log in again. The cookie file, which is a component of the
cookie jar being passed as argument to fetchIndexPage may look like the following after the program
has logged in.

#LWP-Cookies-1.0
Set-Cookie3: entitlements="dsa{?tag"; path="/";
domain="www.apnewstracker.com"; path_spec;

discard; ; version=0
Set-Cookie3: numhits=12; path="/"; domain="www.apnewstracker.com";

path_spec;
discard; ; version=0

Set-Cookie3: sessionid=470092; path="/"; domain="www.apnewstracker.com"; path_spec;
discard; ; version=0

Set-Cookie3: sortval=1; path="/"; domain="www.apnewstracker.com";
path_spec;

discard; ; version=0
Set-Cookie3: userid=6001210; path="/"; domain="www.apnewstracker.com";
path_spec;

discard; ; version=0
Set-Cookie3: usertype=A; path="/"; domain="www.apnewstracker.com";
path_spec;

discard; ; version=0

Each individual line has been broken into two for the purpose of this example since the lines are too long
to be printed.
We do not discuss the syntax of the cookies here. In the cookie file displayed above, each row stands for a
cookie.
After fetching an index page, the program parses the index page using the following call.

&parseIndexPage ($newsItemIndex); #This puts values in %newsIndexHash

Parsing places the news items in the hash %newsIndexHash. Finally, the program goes through each item
in this hash and fetches the text of each news item by following the URL for the news item.

foreach $itemURL (reverse (sort (keys %newsIndexHash))){
$itemText = &fetchNewsItemURL ($ua, $cookie_jar, $itemURL);
if (!$itemText){

print "Could not fetch url: $itemURL\n";
}

} #foreach ends

In the call to fetchNewsItemURL that fetches individual URLs corresponding to individual news items,
the cookies in the stored cookie file are sent to the Web server. This sending of cookies needs to be done
every time when we attempt to retrieve pages from the Web server. This is important. If the appropriate
cookies are not sent with the request, the server rejects the request. In this particular case, since we obtain
the cookies to be sent from the cookie file that got the cookies from the previous page fetched, everything
should be in proper order.
The program has several subroutines: login, fetchIndexPage, parseIndexPage and
fetchNewsItemURL. We briefly look at each one next.
The subroutine login takes a user agent and a cookie jar as the arguments. It makes a POST request using
the following statement.

$request = POST $AP_config::loginURL,
[username => $AP_config::loginName, password => $AP_config::loginPassword];

The request is printed on the screen as a string for verification purposes. The HTTP::Request object needs
to have appropriate cookie headers set. This is done using the following statement.

$cookie_jar -> add_cookie_header ($request);

The contents of the cookies as well as all HTTP interactions taking place between the server and the client
can be seen if we use the LWP::Debug module and import the + symbol. In the current program, the line
that imports this module has been commented out. The line looks like the following and occurs on the top
of the program.

#use LWP::Debug qw (+);

If we want to see the HTTP interactions, this line needs to be uncommented. Seeing what is going back and
forth between the Web server and the Web client is instructive as well as useful for debugging.
The HTTP::Request object $request must have a valid URL attribute before the
add_cookie_headers method can be called. A response is obtained whether the request is successful
or not. The cookies are extracted from the response and the cookie file, the so-called cookie jar, is updated
using the following statement.

$cookie_jar -> extract_cookies ($response);

The subroutine then looks to see if the status code associated with the response indicates success. If so,
it returns the response to the calling program. Otherwise, it prints an error message and returns 0 to the
calling program. Of course, the extracting of cookies can be done after checking the status code.
The fetchIndexPage subroutine also takes the user agent and the cookie jar as arguments. It obtains the
index page for news items using the GET function call of the HTTP::Request::Common module. Cookie
headers are added to the request. From the response, the cookies are extracted and the cookie file is updated
if any of the cookie value is new. If successful, fetchIndexPage returns the content of the page fetched
to the calling program.
The parseIndexPage subroutine parses the contents of the index page fetched earlier. We examine the
HTML of the index page carefully and determine where the individual news items are. Thus, the HTML-
level parsing that takes place in this subroutine is very much dependent on the syntax of the HTML page
under consideration. Web sites are known to change the format of their Web pages frequently, and hence,
the parsing performed here will have to be changed if the structure of the Web page is found to change.
The components obtained for a news item are: $year, $month, $day, $hour, $minute, $file and
$headline. Once the items of news have been obtained, they are joined to form a single string called
$itemData. This string is stored in the global variable %newsIndexHash with the URL of the news item
as its key.
The fetchNewsItemURL subroutine is simple. It gets the user agent, the cookie file, and the URL to fetch
as argument. It creates a GET request, adds the cookie headers to it, and fetches the file.
As indicated earlier, the program described above is the shell of a much more complex program written by
the author and others. The program shows that cookies play an important role in Web programming. A
lot of Web servers use such cookies, which are small pieces of data, to keep track of information about the
clients fetching pages from them. Some cookies are short-lived whereas others are valid for longer periods,
such as days or months. Some cookies are to be discarded by the client whereas others are to be saved for
a certain duration of time. In the cookie file shown earlier, all cookies are to be discarded after the current
session is over.
During a single session, as a Web client goes from page to page within a certain Web site, the server checks
the names and contents of the cookies for various informational bits about the client. Without the appropri-
ate cookie names and the appropriate values, the Web server may refuse to send the Web page requested
to the client. A cookie may indicate if the user is logged in during this session, when the user logged in
the last time, the type of machine and browser being used, etc. Therefore, when we program a Web client
to interact with such a Web server, cookie headers need to be appropriately filled in for each Web page
request. The cookies are saved in a cookie file and usually this file is updated after each interaction with the
Web server. The program discussed above uses cookies fairly extensively and illustrates how they can be
manipulated. More detailed discussion on cookies can found on a Web site such as www.netscape.com
and www.cookiecentral.com.

Figure 9.7: A Search Form at www.fatbrain.com

9.4 Handling Redirected Web Pages

A problem that a Web client program sometimes faces is that when it requests a page from a Web server,
the server responds saying that the page has moved, or something similar. It is possible for Web servers
to redirect requests to pages. To find out if such a situation occurs in a specific case, one has to look at the
result that comes back from the Web server when a program attempts to fill in a form and get response from
a server. Let us take one specific example. Our goal is to find the price of a book at the www.fatbrain.com
site given the ISBN number of a book. We start by going to the www.fatbrain.com site and follow the
links to a form that allows advanced search for books. The form can be found in the page with the URL
http://www1.fatbrain.com/search/AdvancedSearch.asp. The form is shown in Figure 9.7.
We look at the HTML source for this page and see that the ACTION attribute for the form specifies the URL
for the program that handles the form data as
http://www1.fatbrain.com/asp/Search/SearchResults.asp.
Also, the form uses the POST method for transmission of data from the client to the server. Based on this
knowledge, we write the following program.

Program 9.14

#!/usr/bin/perl
#file fatISBN.notwork.pl
use strict;
use HTTP::Response;
use HTTP::Request;
use LWP::UserAgent;
use HTTP::Cookies;
use HTTP::Request::Common;
use LWP::Debug qw(+);

my ($cookie_jar, $ua, $response,$url);
my ($content, $req);
my $searchFor = "0534934056";
$cookie_jar = HTTP::Cookies->new(file=>"cookies.dat");
$ua = LWP::UserAgent->new();
$ua->agent ("Mozilla/5.0");
$ua->timeout(600);

$req = POST "http://www1.fatbrain.com/asp/Search/SearchResults.asp",
[SearchFunction => "reg",
VM => "c",
RegAction => "t",
ISBN => $searchFor];

$cookie_jar->add_cookie_header($req);
$response = $ua->request($req);
$cookie_jar->extract_cookies($response);
$content = $response->content();
print "****content = $content\n";

This program seems adequate for the purpose at hand. Please note that we are using the LWP::Debug
module and imported the + symbol to see on the screen the HTTP interactions taking place. When we
run this program, we find an unexpected error saying that the page has moved permanently. The HTTP
interactions when this program is run and the program’s output to the screen are given below. Some lines
have been broken into two so that they can be printed without crossing into the margin of the paper.

LWP::UserAgent::new: ()
HTTP::Cookies::add_cookie_header: Checking www1.fatbrain.com for cookies
HTTP::Cookies::add_cookie_header: Checking .fatbrain.com for cookies
LWP::UserAgent::request: ()
LWP::UserAgent::simple_request:

POST http://www1.fatbrain.com/asp/Search/SearchResults.asp
LWP::UserAgent::_need_proxy: Not proxied
LWP::Protocol::http::request: ()
LWP::Protocol::http::request: POST /asp/Search/SearchResults.asp HTTP/1.0
Host: www1.fatbrain.com
User-Agent: Mozilla/5.0
Content-Length: 51
Content-Type: application/x-www-form-urlencoded

LWP::Protocol::http::request: POST /asp/Search/SearchResults.asp HTTP/1.0
Host: www1.fatbrain.com
User-Agent: Mozilla/5.0
Content-Length: 51
Content-Type: application/x-www-form-urlencoded

LWP::Protocol::http::request: reading response
LWP::Protocol::http::request: HTTP/1.1 301 Moved
Server: Microsoft-IIS/4.0
Date: Thu, 17 May 2001 22:21:18 GMT
Location: http://www1.fatbrain.com/search/SearchResults.asp?
Connection: Keep-Alive
Content-Length: 0
Content-Type: text/html

Cache-control: private

LWP::Protocol::http::request: HTTP/1.1 301 Moved
LWP::UserAgent::request: Simple response: Moved Permanently
****content =

If we carefully look at toward the bottom of this HTTP interaction, we see that it says that the page that we
have requested, i.e., the program that is supposed to service the form request has Moved Permanently.
In other words, it is possible that the Web server has redirected the request to another URL. To find out
where the redirection is taking us, we need to go the Web browser and perform an ISBN search. Assume
we fill in the ISBN number box with the valid ISBN number 0534934056 and then click on the Search
Now submit button. When the response comes back from the server, we look at the Location box of the
browser and see that the URL that is visible is not
http://www1.fatbrain.com/asp/Search/SearchResults.asp, but is
http://www1.fatbrain.com/asp/bookinfo/bookinfo.asp?theisbn=0534934056&vm=. Thus,
we see that the form request has been redirected to the second URL from the first. The second URL has
a question mark following it, indicating that the submission of the form at the second URL uses the GET
method for form submission. In addition, there is one attribute isbn with the ISBN number as its value.
The second attribute vm does not have an associated value and thus, seems not useful to the current en-
deavor. Therefore, we try to write a program that mimics the behavior of the client and the server and thus,
sends a GET request to second URL. This new request is given the ISBN as an argument. In addition, this
request is given the cookies that came as response to the POST request made to the first URL. Finally, when
the response comes back from the second HTTP request, we parse the page and obtain the price of the book.
The complete program where we take care of redirection is given below.

Program 9.15

#!/usr/bin/perl
#file fatISBN.pl
use strict;
use HTTP::Response;
use HTTP::Request;
use LWP::UserAgent;
use HTTP::Cookies;
use HTTP::Request::Common;

##############fatbrain.com starts##########################
my ($cookie_jar, $ua, $response, $sessid,$url);

my ($content, $req);
my $searchFor = "0534934056";
$cookie_jar = HTTP::Cookies->new(file=>"cookies.dat");
$ua = LWP::UserAgent->new();
$ua->agent ("Mozilla/5.0");
$ua->timeout(600);

$req = POST "http://www1.fatbrain.com/asp/Search/SearchResults.asp",
[SearchFunction => "reg",
VM => "c",
RegAction => "t",
ISBN => $searchFor];

$cookie_jar->add_cookie_header($req);

$response = $ua->request($req);
$cookie_jar->extract_cookies($response);
$content = $response->content();
$url =
"http://www1.fatbrain.com/asp/bookinfo/bookinfo.asp?theisbn=$searchFor";

$req = HTTP::Request->new(GET=>$url);
$cookie_jar->add_cookie_header($req);
$response = $ua->request($req);
$cookie_jar->extract_cookies($response);
$content = $response->content();
#print $content;

#Make the assumption that name, author, etc., are known for the book
my ($price) = ($content =˜ m#Online Price:.+?\$(.+?)<#si);
print $price;

If we use the LWP::Debug module and import the + symbol, we see HTTP communication beyond what
we presented earlier. The additional communication is given below. We have broken some lines so that
they can be printed on the available space.

TTP::Cookies::add_cookie_header: Checking www1.fatbrain.com for cookies
HTTP::Cookies::add_cookie_header: Checking .fatbrain.com for cookies
LWP::UserAgent::request: ()
LWP::UserAgent::simple_request:

GET http://www1.fatbrain.com/asp/bookinfo/bookinfo.asp?theisbn=0534934056
LWP::UserAgent::_need_proxy: Not proxied
LWP::Protocol::http::request: ()
LWP::Protocol::http::request:

GET /asp/bookinfo/bookinfo.asp?theisbn=0534934056 HTTP/1.0
Host: www1.fatbrain.com
User-Agent: Mozilla/5.0

LWP::Protocol::http::request: reading response
LWP::Protocol::http::request: HTTP/1.1 200 OK
Server: Microsoft-IIS/4.0
Date: Thu, 17 May 2001 23:06:05 GMT
Content-Type: text/html
Set-Cookie: Jar=BID=0478B379EF7D2A06;

expires=Mon, 01-Jan-2024 05:00:00 GMT; domain=.fatbrain.com; path=/
Cache-control: private

LWP::Protocol::http::request: HTTP/1.1 200 OK
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 411 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1105 bytes

LWP::Protocol::collect: read 1459 bytes
LWP::Protocol::collect: read 1502 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 1199 bytes
LWP::Protocol::collect: read 1504 bytes
LWP::Protocol::collect: read 1460 bytes
LWP::Protocol::collect: read 95 bytes
LWP::UserAgent::request: Simple response: OK
HTTP::Cookies::extract_cookies: Set cookie Jar => BID=0478B379EF7D2A06

This interaction shows that the second HTTP request is successful and comes back with a response code
of 200 and a response string of OK. The program prints the price of the book we are looking at as 127.95
dollars. Commenting out the use LWP::Debug line on the top of the program causes the program to
suppress the HTTP interaction strings to be printed on the screen, but just print the price.
This program shows that to write successful Web client programs, it is sometimes necessary to peform care-
ful detective activities, cautiously monitor the HTTP interactions, and then find ways to solve the problems
that arise in a straight-forward program by mimicing the behavior of a real browser and its counterpart
Web server in such situations. Determining what a Web browser and a Web server do in such problem-
atic situations is crucial in writing a successful Web client program. The manner in which a client such
as Netscape Communicator works in various situations is publicly available. But, finding the appropriate
action for the situation at hand is usually difficult from among mountains of information available at sites
such as www.netscape.com and www.mozilla.com. There is a mailing list to discuss problems that
arise when programming with the LWP modules. One can subscribe to this mailing list by writing to the
address libwww-subscribe@perl.org. One can post a problem, a response or an experience by writing
to libwww@perl.org. This mailing list is read by many experienced LWP programmers, including the
authors of the various modules. This is the best place to get information and have one’s vexing questions
answered.

9.5 Extracting Links from Web Pages Using HTML::LinkExtor

We have seen in Section 7.4.2.2 how we can extract links from a Web page fetched over the network by
parsing the page. Extracting links and fetching the Web pages pointed to by the links either recursively
or in some other fashion is useful for purposes such as creating indices for a Web site with the purpose of
facilitating searching. Hyperlinks that appear on a Web page point to resources of various kinds: HTML
pages, XML pages, graphic files, audio or video clips, CGI programs, etc. To be able to extract links of a
type we are interested in, or extracting all links from the various HTML attributes they can appear in, is, in
general, time consuming. There is a Perl module HTML::LinkExtor that can be used for this purpose.
The HTML::LinkExtormodule can be used to parse HTML pages fairly easily for extracting links. Below,
we present a program that finds the so-called dead or stale links in a Web site. It starts with a URL given
to it and traverses the site recursively, and examines each traversed page in order to collect stale or dead
URLs.

Program 9.16

#!/usr/bin/perl
#linkExtract5.pl

use strict;
use LWP::UserAgent;
use HTML::LinkExtor;

use URI; #Needed to absolutize a URL, if necessary

$" = "\n";
my $MAXURLCOUNT = 100; #The max no of unique URLs to look at
my $COUNT = 0;
my $ua;
my %TRAVERSED; #Keeps track of all traversed URLs
#Note the / at the end is needed
my $STARTURL = "http://www.cs.uccs.edu/˜kalita/";
#Used to see if a link is inside the domain represented by $BASEURL or outside.
my $BASEURL = "http://www.cs.uccs.edu/";
my $MIMEEXTS = "(s?html?|xml|asp|pl|css}jpg|gif|pdf)";
my ($domain) = ($BASEURL =˜ m#http://(.+?)/?$#);
my $RECORDFILE = "ERRORS.$domain.txt";

#Need a UserAgent to be the client
$ua = new LWP::UserAgent;
open OUT, ">$RECORDFILE";
extractLinks ($STARTURL, $BASEURL);
close OUT;

#####################
sub extractLinks{

my ($url, $containingURL) = @_;
exit if $TRAVERSED{$url};
$TRAVERSED{$url}++;
$COUNT = $COUNT + 1;
if ($COUNT > $MAXURLCOUNT){

exit 0;
}
print "Looking at an in-domain URL #$COUNT: $url\n";

#Make the parser. Can give 0, 1 or 2 args. The first is an optional sub to
#process the urls. the second is used to absolutize any relative URLs
#that may occur.
#Not giving any args here. Absolutization is done separately.
my $p = HTML::LinkExtor -> new ();
#Request document using HTTP
my $res = $ua->request(HTTP::Request->new(GET => $url));
if (!($res -> is_success)){
print OUT
"Stale URL: $url\nContaining URL: $containingURL\nHTTP Message: ",

$res->message, "\n\n";
return;
}
#We have the contents of the file now
my $file = $res -> content;
#This produces an anonymous array with the links
$p -> parse ($file);
my @links = $p->links();
my ($aLinkRef, %linkHash, @linkArray);

foreach $aLinkRef (@links){
my ($tag, $attr, $theUrl) = @$aLinkRef;
#Absolutizing done below doesn’t work if a directory-level
#URL doesn’t have a / at the end.
if ($url !˜ /[.]$MIMEEXTS$/ and $url !˜ m@/$@)

{$url = $url . "/"};
my $theURI = new URI ($theUrl);

$theURI = $theURI->abs($url);
$linkHash{$theURI}++;

}
@linkArray = sort (keys %linkHash);
my $newURL;
foreach $newURL (@linkArray){

next if $TRAVERSED{$newURL};
if ($newURL =˜ m/ˆ$BASEURL/){

extractLinks ($newURL, $url);
}

else {
checkLink ($newURL, $url);

}
}

}

##################
sub checkLink{

my ($url, $containingURL) = @_;
exit if $TRAVERSED{$url};
$TRAVERSED{$url}++;
$COUNT = $COUNT + 1;
if ($COUNT > $MAXURLCOUNT){

exit 0;
}
print "Looking at an out-of-domain URL #$COUNT: $url\n";
#Request document using HTTP
my $res = $ua->request(HTTP::Request->new(GET => $url));

#$res-> code gives the response code
#$res-> is_success is a boolean
#$res -> message returns the error message
if (!($res -> is_success)){

print OUT
"Stale URL: $url\nContaining URL: $containingURL\nHTTP Message: ",
$res->message, "\n\n";

return;
}

}

The program uses the LWP::UserAgent, HTML::LinkExtor and URI modules. The value of
$MAXURLCOUNT specifies the maximum number of unique URLs the program examines. The program has

459



On Perl CHAPTER 9. ON WEB CLIENT PROGRAMMING

two URLs, $STARTURL and $BASEURL. $STARTURL specifies the URL where the recursive examination of
URLs start. $BASEURL is the base URL for the $STARTURL. Clearly, the base URL can be easily obtained by
parsing the $STARTURL and keeping the portion of the URL from the beginning till the end of the server
name. The program culls out the domain name from $BASEURL The program looks for stale URLs and
records the ones it finds in the file $RECORDFILE. We define a stale URL as one that cannot be retrieved
from a Web server during the default timeout period of the user agent. Most frequently, a stale URL corre-
sponds to a page that no longer exists. The recursive extraction of links is done by making a call to the first
extractLinks subroutine. The call is given below.

extractLinks ($STARTURL, $BASEURL);

The subroutine extractLinks takes two arguments: the URL to start searching from, and the base URL of
the starting page. The subroutine is recursive and exits when the count of unique URLs examined exceeds
$MAXURLCOUNT, a global variable. The subroutine starts by examining if the URL $url, to be fetched and
analyzed, has already been examined and hence occurs in the global hash %TRAVERSED. Using a hash to
keep track of examined URLs, we can easily ensure that the URLs examined by the program are unique.
If the URL is not a key in %TRAVERSED, our program enters it in %TRAVERSED before analyzing it. We
examine at most $MAXURLCOUNT URLs.
The subroutine associates a parser with the scalar variable $p. The parse is associated with the following
statement of code.

my $p = HTML::LinkExtor -> new ();

The constructor new for HTML::LinkExtor can take zero, one or two arguments; The first argument, if
given, is reference to a so-called callback subroutine. If a callback subroutine is provided, it is called once
automatically for each link found. If a callback subroutine is not provided, as is the case here, the links are
collected internally and can be retrieved by calling the links method of the parser. The second argument,
if given, is used to absolutize any relative URLs that are obtained from the analyzed Web pages. We do not
specify either of the two arguments in our call. When necessary, we absolutize relative URLs on our own,
as discussed later.
A GET request is created for the Web page addressed by $url and the request sent to the Web server. The
response from the Web server is called $res. If the success code associated with $res indicates that the
GET request is successful, the subroutine stores the content of the response in the variable $file. If the GET
request is unsuccessful, the URL is recorded as stale or dead. Next, the subroutine calls the parse method
of the Link::Extor class on the parser object $p using the argument $file.

$p -> parse ($file);

parse is a method of the HTML::Parse class that we do not discuss here. The class HTML::LinkExtor
is a subclass of HTML::Parse and hence, inherits the parse method from HTML::Parse. The URLs
specified in the HTML file are extracted by the following call.

my @links = $p->links();

The HTML::LinkExtor parse $p stores the links internally in the program. The links() method returns
these links in a list. Here, the list is called @links. @links contains a list of references to links.
The first foreach loop that follows gets each reference to a link and then processes it. A link stored by
the HTML::LinkExtor parser contains three parts: a tag, an attribute, and the URL. For example, if in the
HTML file, the tag is

<A HREF="http://www.cs.uccs.edu/˜kalita">

the tag is A, the attribute is HREF and the URL is http://www.cs.uccs.edu/˜kalita. If the HTML file
contains

<IMG SRC="http://www.cs.uccs.edu/˜kalita/jk1.jpg">

the tag is IMG, the attribute is SRC and the URL is
http://www.cs.uccs.edu/˜kalita/jk1.jpg.
Before we collect the URLs specified in a page, we need to absolutize any relative URLs that may occur in
the page. A relative URL is one that does not start with the protocol string such as http or ftp. Thus, if
we have a relative URL garden-of-the-gods.jpg, and it occurs in the page with absolute URL
http://www.cs.uccs.edu/˜kalita, it is absolutized as
http://www.cs.uccs.edu/˜kalita/garden-of-the-gods.jpg. To absolutize a URL, we use the
abs method in the URI module. The absolutization steps are given below.

#Absolutizing done below doesn’t work if a directory-level
#URL doesn’t have a / at the end.
if ($url !˜ /[.]$MIMEEXTS$/ and $url !˜ m@/$@)

{$url = $url . "/"};
my $theURI = new URI ($theUrl);
$theURI = $theURI->abs($url);

The abs() method of the URI class does not return the correct absolute URL for an argument if the URL
with respect to which absolutization is performed does not end with a /. To take care of this anomalous
situation, we use the global variable $MIMEEXTS that gives a non-exhaustive list of extensions used by non-
directory files, in the form of a regular expression. If the URL $url being analyzed does not end with one
of the file extensions listed in $MIMEEXTS, we assume that $url refers to a directory. If it is a directory, and
does not have a trailing /, we append a / at the end. The next step in absolutization is to take the current
individual URL $theURL found in the page with address $url, and create a URI object out of it. To finish
the absolutization process, the abs method of the URI class is called on $theURI with the containing URL
$url as an argument. If a URL is already an absolute form, it does not change. Otherwise, it is transformed
syntactically to make it absolute.
The first foreach loop goes over every element of @links and takes the URL out of the current link
data structure and stores it in %linkHash. We use a hash so that there is o duplication of the URLs col-
lected. That is, if a URL occurs several times in a Web page, we register it only once. A new array called
@linkArray is created containing the unique URLs that are collected, i.e., the keys of %linkHash.
The extractLinks subroutine has a second foreach loop. In this loop that iterates over every unique
and unexamined URL collected, the subroutine extractLinks is recursively called if the URL happens
to be in the domain we are exploring. Such in-domain link sare examined for staleness in extractLinks.
If the URL is from another domain, the subroutine checkLink is called to see if the URL is stale or dead.
URLs pointing to locations outside the focussed site are not explored recursively.
The checkLink subroutine is called with two arguments: $url, the link to check for staleness and the
URL in which the first URL was mentioned. The subroutine makes a GET request to retrieve $url. If this
request is not successful, the subroutine prints a message saying which URL timed out, and is thus, stale.
The subroutine also prints the URL of the page that contains a reference to the stale URL. The program
prints the URLs it examines on the screen. A partial printout of the screen is given below. Lines have been
broken to fit the printed page.

Looking at an in-domain URL #1: http://www.cs.uccs.edu/˜kalita/
Looking at an out-of-domain URL #2: http://www.assam.org
Looking at an out-of-domain URL #3: http://www.assamcompany.com
Looking at an out-of-domain URL #4: http://www.autoindia.com
Looking at an in-domain URL #5:

http://www.cs.uccs.edu/cgi-bin/jkkalita/counter.pl?counter=kalita-index-page
Looking at an in-domain URL #6:

http://www.cs.uccs.edu/˜kalita/accesswatch/accesswatch-1.32/index.html
Looking at an in-domain URL #7: http://www.cs.uccs.edu/cgi-bin/jkkalita/counter.pl

Looking at an in-domain URL #8: http://www.cs.uccs.edu/˜kalita
Looking at an in-domain URL #9: http://www.cs.uccs.edu/˜kalita/college.gif
Looking at an in-domain URL #10: http://www.cs.uccs.edu/˜kalita/cultural.html
Looking at an out-of-domain URL #11: http://www.amnesty.org

The program stores the list of stale URLs in the file ERRORS.$domain.txt. In this specific case, $domain
has the value www.cs.uccs.edu. Therefore, the record file is called ERRORS.www.cs.uccs.edu. A
partial content of this file after the program is run is given below.

Stale URL: http://www.acsu.buffalo.edu/˜talukdar/assam/humanrightsassam.html
Containing URL: http://www.cs.uccs.edu/˜kalita/assam/human-rights-violations.html
HTTP Message: Not Found

Stale URL:
http://www.cs.uccs.edu/cgi-bin/jkkalita/access_counter.pl.old?counter=human-rights

Containing URL: http://www.cs.uccs.edu/˜kalita/assam/human-rights-violations.html
HTTP Message: Not Found

Stale URL:
http://193.135.156.15/tbs/doc.nsf/c12561460043cb8a4125611e00445ea9/
f2261dd9e000fbe4802565090051a509?OpenDocument

Containing URL: http://www.cs.uccs.edu/˜kalita/assam/human-rights/ajit-bhuyan.html
HTTP Message: Can’t connect to 193.135.156.15:80 (Timeout)

Stale URL: http://www.hri.ca/partners/sahrdc/armed/toc.shtml
Containing URL: http://www.cs.uccs.edu/˜kalita/assam/human-rights/ajit-bhuyan.html#soe/
HTTP Message: Object Not Found

Stale URL: http://www.hri.ca/partners/sahrdc/india/detention.shtml
Containing URL: http://www.cs.uccs.edu/˜kalita/assam/human-rights/ajit-bhuyan.html#soe/
HTTP Message: Object Not Found

Stale URL: http://vag.vrml.org/
Containing URL: http://www.cs.uccs.edu/˜kalita/2000-CS301-roster.html
HTTP Message: Can’t connect to vag.vrml.org:80 (Bad hostname ’vag.vrml.org’)

We note that, some of the URLs are recorded as stale because the URL cannot be accessed within the default
timeout period of the user agent $ua. It is possible to increase the timeout period by using the timeout
method of the LWP::UserAgent class.

9.6 Exercises

1. (Medium to Hard: E-Commerce)
Write a program that compares the prices of books from five different Web bookstores. First, present
an HTML page where a visitor can search for books by title words, author name or ISBN number. Your
program should search the five stores, collate the results, and present them in an easily readable
format. What are some of the problems you face? How do you solve them? How can you make your
program more efficient, more user-friendly and more robust? Discuss ideas regarding these issues.

2. (Hard: E-Commerce, Can be a long-term project)
Write a program that compares prices of electronic goods such as cameras from three to five different
e-commerce sites.

3. (Hard: E-Commerce, Long-term project)

Write a program that compares the prices of airlines tickets. This program is going to be harder to
write than the previous programs. This is mainly because an itinerary may consist of several legs.

4. (Hard: E-Commerce, Long-term project)
Write a program that compares the prices of automobiles from a few dealers. This is going to be
difficult in general because of the plethora of options that manufacturers usually provide for cars.

5. (Hard: Meta-Search Engine, Long-term project)
This problem instructs you to write a meta-search engine. You have a Web page where a user can
search for keywords, just like a commercial search engine such as Google or Lycos. Your CGI program
connects to three or four commercial search engines, performs searches on their sites, collates the
results, and prints the results to the browser. At a minimum, remove duplicates, and present the
information consistently. Different search engines usually return the search results using different
formats. What are some of the problems you face?

6. (Easy: Fetching Web Page, Research)
Write a program that fetches a Web page, given its URL, only if it has been modified from the last time
it was fetched. Run the program from time to time.

7. (Easy to Medium: Monitoring Web Site)
Write a program that detects changes in certain Web page of interest to you. It monitors the page
every few hours. If it detects changes, it alerts you with a mail message. Such a program can be of
value to the programs you are asked to write in the previous problems. This is because you may
spend a lot of time writing your code to perform searches on Web sites to make your programs work.
However, the format in which Web sites return search results can change frequently. Such changes
can reduce the usefulness of your program at any time. Therefore, a program that monitors to see if
the format in which results are entered has changed at a certain Web site can alert you so that you can
make the appropriate changes in your own program.

8. (Medium: Web-site Mirroring)
Write a program that mirrors a Web site locally on your machine. This program copies every directory
recursively to your machine. However, it does not copy everything. It runs from time to time, and
copies only those files and directories that have changed. It copies new files and directories as well.
What are some problems in writing such a program so that it is efficient? Implement some of your
ideas.

9. (Medium: HTML Forms, Research)
There is a Perl module called HTML::Forms that can parse out a Web page and capture the forms in
it. Install this module if you do not have it already. Use this module to capture a form, fill it, and
submit it. Rewrite the programs discussed in the text of this Chapter and the problems here, so that
they use HTML::Forms.

10. (Medium to Hard: Authentication, Cryptography, Research)
Many Web sites require a client to authenticate with a name and a password. Perl provides modules
for doing such authentication. Research into these modules. Write a program that fetches a Web page
that requires authentication.

11. (Easy: Secure Web Sites, Research)
There are Web sites which are secure. Their URLs start with https:// instead of http://. These
sites use encryption to ensure that the data transferred back and forth between the client and the
server is encoded so that interceptors cannot read them. Research how a Perl program can obtain a
Web page from such a secure server. Write a program that obtains such a page.

12. (Hard: Movie Database, Can be a long-term project)
Write a program that searches for a movie’s review from a site such as www.imdb.com (The Interna-
tional Movie Database). Write code for an HTML form that allows one to search for a movie with a
keyword. The CGI program associated with this form actually performs searches at the IMDB or a
similar site. It then obtains the names of the movies that the site returns. Your program further obtains
reviews of these movies from the site, if necessary by performing traversal of additional hyperlinks.
The CGI program you write presents the name of each movie followed by its review.

13. (Medium to Hard: Text Processing, Vector Computation, Can be a long-term research project)
Write a program that is given a URL as command-line argument. It fetches the URL. Assume it is a
text file. Remove all the HTML tags from the page. Remove commonly used words such as it, is, some,
etc. Count the frequencies of occurrence of the words in the file.
Now, you are given the URL of another Web page. Repeat the above steps for the second file as well.
Make a single sorted list of the words in the two files. Assign a numeric ID to each word. Form a
frequency vector, one per URL. You will create two frequency vectors. The vector for a file should
contain the frequencies of the words in the file, in order of the word number. Words that do not occur
in a file have the frequncy of zero for that file. Find a way to normalize this vector so that the values do
not grow unbounded, i.e., become large. In addition, it is easier to do computation with normalized
numbers.
Given these two vectors, we can perform a computation called the cosine computation to find how
similar the two files are. The cosine computation treats the finds the cosine of the angle between the
two vectors. Although the dimension of the two vectors is large, we can still find the angle between
the two vectors by computing its cosine. Find out the cosine of the angle between the two vectors for
the two URLs. It will tell us how similar are the two Web pages that we started with. A small value
of the cosine means that the angle between the two vectors is small.
Now, we will extend the program. We are still given the first URL to start with. It is the base page
with which we will compare all the other pages. The program is also given another URL. It traverses
the Web site represented by the second URL recursively. It finds the similarity of each page with the
base page. It prints the similarity numbers in a table. It also sorts the pages by the similarity values
in ascending order.
A program such as this can be used to traverse the Web and automatically classify pages according to
whether they are similar to a base page or not. The base page’s vector does not have to correspond to
a real page, but could be the representation of a class of pages. For example, a vector can represent
the characteristic word frequencies for a class such as news. A program like the one you have written
can then crawl the Web and automatically find Web sites that are news related.

